

astal Systems Station, Dahlgren Division
val Surface Warfare Center
ma City, Florida 32407-5000



2

AD-A248 597



TECHNICAL MEMORANDUM
CSS TM 596-92

FEBRUARY 1992

**STATISTICS OF NARROWBAND WHITE NOISE
DERIVED FROM CLIPPED BROADBAND
WHITE NOISE**

W. M. WYNN



Approved for public release; distribution is unlimited

92-09301



24 92 4 10 085



**Coastal Systems Station, Dahlgren Division
Naval Surface Warfare Center**

PANAMA CITY, FLORIDA 32407-5000

CAPT D. P. FITCH, USN
Commanding Officer

MR. TED C. BUCKLEY
Executive Director

ADMINISTRATIVE INFORMATION

This work is a byproduct of a long-term analysis of broadband countermeasure signals. It is not identified with any particular project or program, but addresses a question of general concern in this area.

Released by
D. P. Skinner, Head
Coastal Technology Department

Under authority of
T. C. Buckley
Executive Director

REPORT DOCUMENTATION PAGEForm Approved
OMB No. 0704-0188

Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.

1. AGENCY USE ONLY (Leave blank)

2. REPORT DATE

February 1992

3. REPORT TYPE AND DATES COVERED

4. TITLE AND SUBTITLE

Statistics of Narrowband White Noise Derived from Clipped Broadband White Noise

5. FUNDING NUMBERS

6. AUTHOR(S)

W. M. Wynn

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)

Coastal Systems Station
Code N1130
Panama City, Florida 32407-50008. PERFORMING ORGANIZATION
REPORT NUMBER

CSS TM 596-92

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)

10. SPONSORING/MONITORING
AGENCY REPORT NUMBER

11. SUPPLEMENTARY NOTES

12a. DISTRIBUTION/AVAILABILITY STATEMENT

Approved for public release, distribution is unlimited

12b. DISTRIBUTION CODE

13. ABSTRACT (Maximum 200 words)

Broadband white Gaussian noise is modelled as a series of sine and cosine functions of discrete frequencies up to a maximum frequency. The series coefficients are Gaussian with uniform variance across the frequency band. For a given sample, the Nyquist-spaced values are generated for a full period of the lowest frequency sinusoid and a peak factor is imposed, clipping the values. The resulting signal is discrete-Fourier transformed via the fast Fourier transform, hard filtered at a prescribed bandwidth, and then inverse transformed to give a set of values for the narrow band signal sample, and these values are used to update a histogram. When a sufficient number of signal samples are generated, the resulting distribution is fitted to a Gaussian distribution by a least-squares technique. It is found that the resulting distributions are noticeably non-Gaussian down to a bandwidth ratio of 1:4, below which the distribution appears to be Gaussian with a reduced variance depending on the peak factor. For a bandwidth ratio of 1:40, it is found that peak factors as low as 1.7 or even 1.5 have little effect on the narrowband distribution.

14. SUBJECT TERMS

White noise; Gaussian noise; Broadband signal; Lease square; Rocky Mountain;
Histogram; Narrowband

15. NUMBER OF PAGES

13

16. PRICE CODE

17. SECURITY CLASSIFICATION
OF REPORT

UNCLASSIFIED

18. SECURITY CLASSIFICATION
OF THIS PAGE

UNCLASSIFIED

19. SECURITY CLASSIFICATION
OF ABSTRACT

UNCLASSIFIED

20. LIMITATION OF ABSTRACT

SAR

CONTENTS

	<u>Page No.</u>
INTRODUCTION	1
THE SIGNAL MODEL	1
NARROWBAND STATISTICS	2
APPENDIX A- LEAST-SQUARES FITTING OF A GAUSSIAN DISTRIBUTION TO A HISTOGRAM	A-1
APPENDIX B- ROCKY MOUNTAIN BASIC 3.0 CODE FOR SIGNAL SAMPLE GENERATION, FILTERING, AND HISTO- GRAM FORMATION	B-1

ILLUSTRATIONS

<u>Figure No.</u>		<u>Page No.</u>
1	Gaussian Fits for a Bandwidth Ratio of 1:40 for Various Peak Factors	4
2	Histograms and Gaussian Fits Compared for a Peak Factor of 1.7 for Various Bandwidth Ratios	5

Accession For	
NTIS GRA&I	<input checked="" type="checkbox"/>
DTIC TAB	<input type="checkbox"/>
Unannounced	<input type="checkbox"/>
Justification	
By	
Distribution	
Availability Codes	
Dist	Avail and/or Special
A-1	

INTRODUCTION

In many countermeasure applications, it is necessary to broadcast broadband white noise to influence a device that has a relatively narrowband processor of unknown center frequency. Due to practical limitations, the excursions of the power supply generating the broadband noise are hard limited by a peak factor. An important operational question then is: to what extent does this limit the excursions of a narrowband-filtered output of the power supply?

In this note, the broadband white noise is modelled as a finite sum of discrete-frequency sinusoids, with Gaussian-distributed amplitudes derived from a uniform distribution. The resulting time-domain signal then is hard limited, and the discrete Fourier transform is applied (exact for this type of signal) and executed via the fast Fourier transform. The resulting transform amplitudes then are hard filtered for a particular choice of bandwidth, and the inverse discrete Fourier transform is applied. Time samples of the resulting narrowband time signal then are used to update a histogram of values, the amplitudes of the broadband time signal then are re-seeded, and the process repeated until a sufficient number of signal samples are collected.

The resulting histogram is used to construct a least-squares fit to a Gaussian distribution, and the resulting variance is compared to that for the unclipped case.

THE SIGNAL MODEL

The broadband signal has the form

$$B(n\Delta t) = \sum_{m=1}^{N/2} [a_m \cos(2\pi n m \Delta t \Delta f) + b_m \sin(2\pi n m \Delta t \Delta f)] \quad (1)$$

where $\Delta f = 1/T$ and $\Delta t = 1/2F$ with T the period of the lowest frequency sinusoid, and F the highest frequency present. We will choose N to be a power of 2, and note that $1 \leq n \leq N$, $1 \leq m \leq N/2$, and $\Delta t \Delta f = 1/N$.

The coefficients are chosen using the following algorithm:

$$U_1 \in U[0, 1] \quad , \quad U_2 \in U[0, 1] \quad (2)$$

$$V_1 = 2U_1 - 1 \quad , \quad V_2 = 2U_2 - 1 \quad (3)$$

that is, U_1 and U_2 are chosen from a distribution uniform on the interval $[0, 1]$, and are used to generate a distribution uniform on $[-1, 1]$, of which V_1 and V_2 are samples (most computers generate the distribution uniform on $[0, 1]$ by means of a pseudo-random number generator).

Define $S = V_1^2 + V_2^2$. If $S > 1$ then select new samples U_1 and U_2 from the uniform distribution and repeat the process. The resulting variable S is distributed uniformly on $[0, 1]$. Next, define

$$X_1 = V_1 \sqrt{\frac{-2\sigma^2 \ln S}{S}} \quad (4)$$

and

$$X_2 = V_2 \sqrt{\frac{-2\sigma^2 \ln S}{S}}. \quad (5)$$

The resulting variables X_1 and X_2 have Gaussian distributions with zero mean and variance σ .

For a given broadband signal sample, a pseudo-random sequence is seeded, and the coefficients a_m and b_m are selected using the above procedure for $1 \leq m \leq N/2$. When a new signal sample is constructed, the sequence is re-seeded.

When a peak factor P is specified, the clipped signal has the form

$$C(n\Delta t) = B(n\Delta t), B < \sigma P, C(n\Delta t) = \sigma P \text{ otherwise.} \quad (6)$$

The discrete Fourier transform of the clipped signal is given by

$$D(l\Delta f) = \sum_{n=1}^N C(n\Delta t) e^{-2\pi i l n / N} \quad (7)$$

with the inverse transform given by

$$C(n\Delta t) = \frac{1}{N} \sum_{l=1}^N D(l\Delta f) e^{2\pi i l n / N}. \quad (8)$$

The validity of this transform pair can be established by means of the identity

$$\sum_{m=1}^N e^{2\pi i (n-k)m/N} = \sum_{l=-\infty}^{\infty} \delta_{n,k+lN}. \quad (9)$$

NARROWBAND STATISTICS

The discrete Fourier transform and inverse transform can be executed via the fast Fourier transform algorithm. A subroutine for performing this algorithm is included as part of the code given in Appendix B. The code is written in Hewlett-Packard's Rocky Mountain Basic Version 3.0. The main body of code constructs a histogram for a particular bandwidth and peak factor choice. For a particular broadband signal sample, the time signal is transformed, and then hard filtered by eliminating the transform coefficients outside the chosen band. In doing this, it is important to observe the redundancy and symmetry of the transform coefficients. In particular, because we are using a real time series as input, the transform coefficients satisfy

$$\text{Re}\{D(l\Delta f)\} = \text{Re}\{D([N-l]\Delta f)\} \quad , \quad l = 1, \dots, N/2 \quad (10)$$

and

$$\text{Im}\{D(l\Delta f)\} = -\text{Im}\{D([N-l]\Delta f)\} \quad , \quad l = 1, \dots, N/2. \quad (11)$$

Consequently, a set of bandpass indices is selected symmetrically about $l = N/2$. Once the filtering is accomplished, the coefficients are inverse transformed to produce the associated time signal. Each time sample is examined and assigned to the appropriate histogram bin. Once this is done, the process is repeated until a sufficient number of samples is accumulated.

For explicit analysis, we have chosen $F = 2048$, and $\Delta f = 4$, so $N = 1024$, $T = 1/4$ and $\Delta t = 1/4096$. The broadband signal is chosen to have $\sigma = 1$, so the amplitudes a_m and b_m are chosen via the above procedure using a σ of $\sqrt{1/512}$. For the filter we consider bandwidth ratios of 1:40, 1:4, 1:2, 3:4, 7:8, 125:128, and 1:1, centered on the frequency 1024. The histogram is constructed as an array of $2M$ elements with element M corresponding to zero amplitude. In some cases, we have used $M = 100$ and in some cases $M = 1000$. It doesn't appear to make much difference in the resulting least-squares Gaussian fit. The matrix index change corresponding to σ for the unclipped case is arbitrarily chosen to be $M/10$ and this number is multiplied by the signal value and divided by the square root of the bandwidth ratio and the result added to M to give the appropriate histogram index. This will give the same distribution for the unclipped case for all bandwidth ratios. Then the effects of the peak factor will appear as departures from a Gaussian distribution, and modifications of the heights and widths of the Gaussian fits.

The distributions and Gaussian fits for a bandwidth ratio of 1:40, for peak factors of 1.7, 1.5, 1.25, 1.0, and 0.5 are shown in Figure 1. Here, and in Figure 2, the abscissa is labelled in units of σ for the unclipped case. The smooth curves are produced by the least-squares fitting procedure described in Appendix A. It is clear from the comparisons given in 1(b)-1(f) that the distributions are Gaussian. The superposition of fits in 1(a) shows that the effect of increasing the peak factor is to reduce the variance and increase the distribution's peak amplitude. It is clear from the curves in 1(a) that imposing a peak factor of 1.7, or even 1.5 has very little practical effect on the resulting narrowband distribution. Note that for peak factors of 1.5, 1.25, and 1.0, we have used $M = 1000$, and have 203 samples, whereas for peak factors of 1.7 and 0.5, we use $M = 100$, and have 146 samples.

To determine when the peak factor does have a significant impact on the shape of the distribution, we start at the other extreme, the distribution for the full bandwidth clipped signal, and reduce the bandwidth in slight increments. In Figure 2, we show the distributions, for peak factor 1.7, for bandwidth ratios 1:1, 125:128, 7:8, 3:4, 1:2, and 1:4. In the 1:1 case shown in 2(a), as expected, the distribution has a delta function behavior at $\pm 1.7\sigma$. In the next case, that for the slight reduction to a 125:128 ratio, shown in 2(b), the distribution is still strongly multimodal. The distribution is still noticeably non-Gaussian for subsequent bandwidth reductions, shown in 2(c)-2(e), but for a ratio of 1:4, shown in 2(f), the distribution is nearly Gaussian with a variance of 92 percent of that of the unclipped case. In these examples, the number of samples varies as follows: 1:1(303), 125:128(177), 7:8(178), 3:4(178), 1:2(223), and 1:4(123).




Figure 1. Gaussian Fits for a Bandwidth Ratio of 1:40, for Peak Factors of 1.7, 1.5, 1.25, 1.0 and 0.5. The Percentage Variances Listed after the Peak Factors are Given Relative to an Uncutted Value of 1. (a) Gaussian Fits Compared, Including the Uncutted Case. (b) 1.7, 96.8%. (c) 1.5, 93.1%. (d) 1.25, 84.9%. (e) 1.0, 75.6%. (f) 0.5, 44.5%.




Figure 2. Histograms and Gaussian Fits Compared for a Peak Factor of 1.7, for Various Bandwidth Ratios: (a) 1:1, (b) 125:128, (c) 7:8, (d) 3:4, (e) 1:2, (f) 1:4. The Distribution is Strongly Non-Gaussian Initially, but Approaches a Gaussian Distribution of Reduced Variance as Bandwidth Decreases.

APPENDIX A

LEAST-SQUARES FITTING OF A GAUSSIAN DISTRIBUTION TO A HISTOGRAM

Given a series of variable and probability values $\{x_i, y_i\}$ where $i = 1, \dots, N$, we would like to establish the parameter σ for the representation

$$y_i = \frac{1}{\sqrt{2\pi}\sigma} e^{-\frac{x_i^2}{2\sigma^2}}. \quad (A1)$$

By taking the natural logarithm of both sides of (Eq. A1), we can produce the equivalent equation

$$u_i = v_i w + \ln w \quad (A2)$$

where

$$u_i = 2 \ln(\sqrt{2\pi} y_i) \quad (A3)$$

$$v_i = -x_i^2 \quad (A4)$$

and

$$w = \frac{1}{\sigma^2}. \quad (A5)$$

The least-square fit is accomplished by numerically solving

$$\frac{d}{dw} (u_i - v_i w - \ln w)^2 = 0 \quad (A6)$$

or, expanding, and incorporating all of the data by averaging,

$$\langle v_i^2 \rangle w + \langle v_i \rangle \ln w - \frac{\langle u_i \rangle}{w} + \frac{\ln w}{w} = \langle u_i v_i \rangle - \langle v_i \rangle. \quad (A8)$$

A subroutine for performing this process is given on the following page.

Rocky Mountain Basic 3.0 Code

```

10 SUB Gaushisfit(X(*),Y(*),Npts,S1,S2,Nmax,S)
20 !
30 !Gaussian fit to y(x) using Npts points to produce variance S
40 !S1 and S2 bound the search interval for S, Nmax is the number
50 !of partition refinements allowed.
60 !
70 OPTION BASE 1
80 DIM U(2000),V(2000)
90 REDIM U(Npts),V(Npts)
100 Lc=LOG(2*PI)
110 Nav=Npts
120 FOR I=1 TO Npts
130 IF ABS(Y(I))<1.E-10 THEN
140 U(I)=0
150 V(I)=0
160 Nav=Nav-1
170 ELSE
180 U(I)=Lc+2*LOG(Y(I))
190 V(I)=-X(I)*X(I)
200 END IF
210 NEXT I
220 A=DOT(V,V)/Nav
230 B=SUM(V)/Nav
240 C=SUM(U)/Nav
250 D=DOT(U,V)/Nav-B
260 Xl=1/S2/S2
270 Xu=1/S1/S1
280 N=0
290 Y0=A*Xl+B*LOG(Xl)-C/Xl+LOG(Xl)/Xl-D
300 Dx=(Xu-Xl)/10
310 FOR Xx=Xl TO Xu+Dx/2 STEP Dx
320 Yy=A*Xx+B*LOG(Xx)-C/Xx+LOG(Xx)/Xx-D
330 IF Yy*Y0<0 THEN
340 N=N+1
350 IF N<Nmax THEN
360 Xl=Xx-Dx
370 Xu=Xx
380 GOTO 290
390 ELSE
400 S=1/SQR(Xx)
410 GOTO 450
420 END IF
430 END IF
440 NEXT Xx
450 SUBEND

```

APPENDIX B

**ROCKY MOUNTAIN BASIC 3.0 CODE FOR SIGNAL SAMPLE
GENERATION, FILTERING, AND HISTOGRAM FORMATION**

```

10  OPTION BASE 1
21  DIM X1(512),X2(512),C(512),S(512),Br(1024),Bi(1024),H(2000)
30  MAT H= (0)
40  Samp=0
50  INPUT "INPUT PEAK FACTOR",P
51  !Peak value is P times the broadband rms value
60  INPUT "FREQUENCY INDEX FOR NARROW BAND FILTER",If1
61  !Midpoint of the digital filter, In units of 4 hertz
70  INPUT "READ IN A PRIOR HISTOGRAM(Y/N)?",Ph$
80  IF Ph$="Y" THEN
90    INPUT "MSUS OF INPUT FILE?",Mh$
100   INPUT "FILE NAME?",H$
110   ASSIGN @Filein TO H$&Mh$
120   ENTER @Filein;P,Samp,H(*)
130   ASSIGN @Filein TO *
140   INPUT "DOES FILE EXIST ON MSI(Y/N)?",Fe$
150   IF Fe$="Y" THEN
160     GOTO 240
170   ELSE
180     GOTO 230
190   END IF
200 ELSE
210   INPUT "NAME OF OUTPUT FILE FOR HISTOGRAM?",H$
220 END IF
230 CREATE BDAT H$,2002,8
240 ASSIGN @File TO H$
241 INPUT "INPUT THE HALF WIDTH OF THE FILTER IN UNITS OF 4 hertz",Dlf
242 !Number of lines retained on each side of the midpoint, Ifil
243 Sig2=1/512
250 !Sigma squared for 4 hertz interval if 1 for 2048 hertz
251 RANDOMIZE
260 !Re-seed for each broadband signal sample
270 FOR I=1 TO 512
280   U1=RND
290   U2=RND
300   V1=2*U1-1
310   V2=2*U2-1
320   S0=V1*V1+V2*V2
330   IF S0>=1 THEN GOTO 280
340   Rt=SQR(-2*Sig2*LOG(S0)/S0)
350   X1(I)=V1*Rt
360   X2(I)=V2*Rt
370 NEXT I
380 Samp=Samp+1
390 PRINT Samp
400 Timesig: !
410 FOR It=1 TO 1024
411   T=(It-1)/4096
420   !Nyquist spacing for 2048 maximum frequency
430   FOR I=1 TO 512
440     C(I)=COS(2*PI*4*I*T)
450     S(I)=SIN(2*PI*4*I*T)
451     !Line spacing is 4 hertz

```

```

460 NEXT I
470 Bi(It)=0
480 Br(It)=DOT(X1,C)+DOT(X2,S)
490 IF ABS(Br(It))>P THEN
500   Br(It)=SGN(Br(It))*P
510   !Clip the signal sample
510 END IF
520 NEXT It
530 CALL Fft(Br(*),Bi(*),1024,10,-1)
531 !Get discrete transform coefficients
540 MAT Br= (1/1024)*Br
550 MAT Bi= (1/1024)*Bi
560 FOR I=1 TO 513-Ifil-Dlf-1
570   Br(I)=0
580   Bi(I)=0
590 NEXT I
600 FOR I=513-Ifil+Dlf+1 TO 513+Ifil-Dlf-1
610   Br(I)=0
620   Bi(I)=0
630 NEXT I
640 FOR I=513+Ifil+Dlf+1 TO 1024
650   Br(I)=0
660   Bi(I)=0
670 NEXT I
671 !Bandpass filter
680 CALL Fft(Br(*),Bi(*),1024,10,-1)
681 !Create the set of time values for this narrowband signal sample
690 FOR I=1 TO 1024
700   Ix=INT(Br(I)*100*SQR(2048/(2*4*Dlf))+1000)
701   !Normalize such that the histogram is independent of the bandwidth
702   !for the unclipped case
710   IF Ix>0 THEN
720     IF Ix<2001 THEN
730       H(Ix)=H(Ix)+1
731       !Upgrade the histogram
740     END IF
750   END IF
760 NEXT I
770 OUTPUT @File;P,Samp,H(*)
771 !Upgrade the stored histogram
780 GOTO 240
781 !Go back and re-seed and do it all again
790 END
800 SUB Fft(Xreal(*),Ximag(*),INTEGER N,Nu,Isign)
810   INTEGER N2,Nu1,K,K1,L,I,K1n2,P
820   N2=N DIV 2
830   Nu1=Nu-1
840   K=0
850   FOR L=1 TO Nu
860     FOR I=1 TO N2
870       CALL Bitrev(P,K DIV INT((2^Nu1)+.001),Nu)
880       Arg=2*PI*P/N
890       C=COS(Arg)

```

```

900   S=SIN(Arg)
910   K1=K+1
920   K1n2=K1+N2
930   Treal=Xreal(K1n2)*C+Ximag(K1n2)*S
940   Timag=Ximag(K1n2)*C-Xreal(K1n2)*S
950   Xreal(K1n2)=Xreal(K1)-Treal
960   Ximag(K1n2)=Ximag(K1)-Timag
970   Xreal(K1)=Xreal(K1)+Treal
980   Ximag(K1)=Ximag(K1)+Timag
990   K=K+1
1000  NEXT I
1010  K=K+N2
1020  IF K<N THEN GOTO 860
1030  K=0
1040  Nu1=Nu1-1
1050  N2=N2 DIV 2
1060  NEXT L
1070  FOR K=1 TO N
1080    CALL Bitrev(I,K-1,Nu)
1090    I=I+1
1100    IF I<=K THEN
1110      Treal=Xreal(K)
1120      Timag=Ximag(K)
1130      Xreal(K)=Xreal(I)
1140      Ximag(K)=Ximag(I)
1150      Xreal(I)=Treal
1160      Ximag(I)=Timag
1170    END IF
1180  NEXT K
1190  IF Isign=1 THEN
1200    FOR K=1 TO N
1210      Ximag(K)=-Ximag(K)
1220    NEXT K
1230  END IF
1240  SUBEND
1250  SUB Bitrev(INTEGER Ibitr,J,Nu)
1260    INTEGER I,J1,J2
1270    J1=J
1280    Ibitr=0
1290    FOR I=1 TO Nu
1300      J2=J1 DIV 2
1310      Ibitr=Ibitr*2+(J1-2*J2)
1320      J1=J2
1330    NEXT I
1340  SUBEND

```


DISTRIBUTION LIST

	<u>Copy No.</u>
Chief of Naval Operations, Navy Department, Washington, DC 20350-2000	
Code NOP-095X	1
Code NOP-098X	2
Code NOP-21T2	3
Code NOP-987B	4
Commander, Naval Air Systems Command, Naval Air Systems Command Headquarters, Washington, DC 20361-0001	
Code NAIR-933, Mr. Barry L. Dillon	5-6
Commander, Naval Sea Systems Command, Naval Sea Systems Command Headquarters, Washington, DC 20362-5101	
Code NSEA 62D1	7
Code NSEA PMS415B	8
Director, Naval Post Graduate School, Monterey, CA 93943	9
Commanding Officer, Naval Technical Intelligence Center, 4301 Suitland Road, Washington, DC 20390-5140	
Attn. Mr. Gerry Batts	10
Officer in Charge, David W. Taylor Naval Ship Research and Development Center, Carderock Laboratory, Bethesda, MD 20084-5000	11
Officer in Charge, David W. Taylor Naval Ship Research and Development Center, Annapolis Laboratory, Annapolis, MD 21402-1198	
Attn. Dr. Bruce Hood	12
Commander, Naval Air Development Center, Warminster, PA 18974-5000	
Attn. Dr. Lloyd Bobb	13
Commander, Naval Surface Weapons Center Detachment, White Oak Laboratory, 10901 New Hampshire Avenue, Silver Spring, MD 20903-5000	
Attn. Dr. John Holmes	14
Attn. Mr. John Stahl	15
Commanding Officer, Naval Surface Weapons Center, Dahlgren Laboratory, Dahlgren VA 22448	16
Commander, Naval Underwater Systems Center, Newport, RI 02841-5047	17
Officer in Charge, Naval Underwater Systems Center, New London Laboratory, New London CT 06320	18
Commander, Naval Ocean Systems Center, San Diego, CA 92152-5000	19
Commander, Naval Weapons Center, China Lake, CA 93555-6001	20
Commanding Officer, Naval Research Laboratory, Washington, DC 20375	21
Director, Defense Advanced Research Projects Agency, 1400 Wilson Blvd, Arlington, VA 22209	22
Administrator, Defense Technical Information Center, Cameron Station, Alexandria, VA 22304-6130	23-24